

Empirical Model: ENSO, Solar, Volcanic Aero, Anthro

http://www.leapcad.com/Climate_Analysis/Empirical_Model_ENSO_Solar_VolcAero_Anthro.xmcd

On time scales of 10 to 50 years (and longer) decadal climate forecasts are difficult to make with general circulation climate models due to their many uncertainties [IPCC, 2007]. We will use the methodology first used by Schonwiese and Bayer in "Some statistical aspects of anthropogenic and natural forced global temperature change", 1995. We will model global temperature by with a multiforced lagged regression by combining ENSO, Volcanic Aerosols, Irradiance, and the effects of Anthropogenic Influence.

Analysis

Using the most recently available characterizations of ENSO, E, volcanic aerosols, V, solar irradiance, S, and anthropogenic influences, A, we perform multiple linear regression analysis to decompose monthly mean surface temperature anomalies since 1980 into four components.

Monthly mean surface temperature anomalies ΔT_{MS} are reconstructed as:

$$\Delta T_{MS}(t) = c_0 + c_E \cdot E(t - \Delta t_E) + c_V \cdot V(t - \Delta t_V) + c_S \cdot S(t - t_S) + c_A \cdot A(t - t_A)$$

Where E, V, S and A are the time series and the lags (in months) are $\Delta t_E = 3$, $\Delta t_V = 6$, and $\Delta t_S = 0$ and $\Delta t_A = 17$ years. The lags are chosen to maximize the proportion of global variability that the statistical model captures and are spatially invariant (although a geographical dependence is expected). The fitted coefficients, c_0, \dots , are obtained by multiple linear regression against the instrumental surface temperature record (HadCRUT3v).

The multivariate ENSO index, E, is a weighted average of the main ENSO features contained in sea-level pressure, surface wind, surface sea and air temperature, and cloudiness [Wolter and Timlin, 1988]. Volcanic aerosols, V, in the stratosphere are compiled by Sato et al. [1993] since 1850, updated from giss.nasa.gov to 1999 and extended to the present with zero values. Although some volcanic activity occurred between 2006 and 2008, it is difficult to calculate the aerosol optical depth because of the lack of direct quantitative space-based observations. Solar irradiance, S, is estimated as the competing effects of sunspots and facular, identified in observations made by space-based radiometers [Lean et al., 2005]. The anthropogenic influence, A, is the Forcing Effect of the concentration (ppm) of CO₂.

Climate Forecasting:

Using global and regional surface temperature responses to the four individual influences parameterized by regression against the observations from 1980 to 2008, we forecast change from 2009 to 2020 by adopting the best estimate of how each influence will change in the future. The anthropogenic forcing in the past 40 years is well represented by a linear trend that we extrapolate into the future.

We assume that future solar irradiance cycles replicate cycle 23, with cycle 24

(See: http://www.leapcad.com/Climate_Analysis/Climate_Data-Proxies_and_Reconstructions.pdf pg. 21) commencing at the beginning of 2009. Although solar activity (as indicated by sunspot numbers) was less in cycle 23 than in cycles 21 and 22, the total irradiance amplitude (near 0.1%) is similar in the three past cycles since it is the net effect of sunspot darkening and facular brightening, both of which are altered by solar activity. Since ENSO fluctuations and volcanic eruptions are not predictable on decadal time scales, we estimate their maximum likely future impact with a scenario that includes a Pinatubo-like eruption with peak impact in 2014 and a super ENSO with maximum impact in 2019, mimicking a similar sequence that occurred from 1992 to 1997 (Figure Stratospheric Optical Depths).

The Data: Temperature, ENSO Index, Volcanic Aerosols, Anthropogenic (CO2 ppm)

HadCrut Temperature and CO2 ppm

<http://www.cru.uea.ac.uk/cru/data/temperature/hadcrut3vgl.txt> Monthly Temp Data 1850 to 2009

Read data from http://www.esrl.noaa.gov/gmd/ccgg/trends/co2_data_mlo.

MLCO2 := READPRN("NOA_Mauna_Loa_Monthly_CO2.TXT")

Date := MLCO2^{<2>} CO2_{ML} := MLCO2^{<4>} TrendCO2 := MLCO2^{<5>} RD := rows(Date)

Get CO2 Trend Line from 1990 to 2010, then project to 2020

Date₁₉₉₀ := submatrix(Date, 383, RD - 1, 0, 0) L_{CO2} := line(Date₁₉₉₀, submatrix(TrendCO2, 383, RD - 1, 0, 0))

Trend_{CO2}(Year) := L_{CO2} + L_{CO2} · Year Co := 280 $\frac{m}{12}$:= 0..12·10 Yr₂₀₂₀_m := 2010 + $\frac{m}{12}$

$$\text{Keeling}(\text{yr}) := 1.054 \cdot 10^{-2} (\text{yr} - 1960)^2 + 9 \cdot 10^{-1} \cdot (\text{yr} - 1960) + 315.5$$

SipleCO2 := READPRN("Friedli Siple CO2 1986.TXT") IceCO2 := READPRN("CO2 Ice Core Data.txt")

<http://www.wasserplanet.becsoft.de/180CO2/CO2tot1812-2007.txt>

Column C: CO2 total 1812-1961 corrected. annual averages from raw data.

HadCrut := READPRN("hadcrut3vgl.txt") rows(HadCrut) = 320 cols(HadCrut) = 14 n := 0..159


HadCrutx := READPRN("hadcrut3glx.txt") rows(HadCrutx) = 160 cols(HadCrutx) = 14 n := 0..159

$$\text{TCrut}_n := \sum_{m=1}^{12} \left(\text{HadCrut}_{2,n,m} \cdot \frac{1}{12} \right) \qquad \text{Time}_{\text{crut}_n} := \text{HadCrut}_{2,n,0}$$

$$\text{TCrut}_n := \text{HadCrutx}_{n,13}$$

$$\text{TCrutPlus} := \overrightarrow{(\Phi(\text{TCrut}) \cdot \text{TCrut})}$$

Global Temp (Bars) & CO2 Levels (Black Line) , To 2020 (Green)




ENSO Index Monthly Data from 1950 to 2009

<http://www.esrl.noaa.gov/psd/people/klaus.wolter/MEI/mei.html> Monthly ENSO Data 1850 to 2009

MEIM := READPRN("MultiVariate ENSO Index.TXT") $R := \text{rows}(\text{MEIM})$ $rr := 0..(R-1)\cdot 12 + 11$

MEIx := submatrix(MEIM, 0, R - 1, 1, 12)

$$\text{MEI}_{rr} := \text{MEIx}_{\text{floor}\left(\frac{rr}{12}\right), \text{mod}(rr, 12)} \quad \text{MEID}_{rr} := \left(\text{MEIM}^{(0)}\right)_{\text{floor}\left(\frac{rr}{12}\right) + \frac{\text{mod}(rr, 12)}{12}}$$




Pacific Decadal Oscillation (PDO) Index from 1900 to 2009

Note: ENSO and PDO are not statistically independent. They have a 47% correlation.

<http://jisao.washington.edu/data/pdo/> Year, Jan to Dec

PDO := READPRN("PDO_Index.dat") $R := \text{rows}(\text{PDO})$ $rr := 0..(R-1)\cdot 12 + 11$

$$\text{PDOI}_{rr} := \text{PDOx}_{\text{floor}\left(\frac{rr}{12}\right), \text{mod}(rr, 12)} \quad \text{PDOY}_{rr} := \left(\text{PDO}^{(0)}\right)_{\text{floor}\left(\frac{rr}{12}\right) + \frac{\text{mod}(rr, 12)}{12}}$$




Stratospheric Volcanic Aerosols

Note: Volcanic Aerosols and ENSO are not statistically independent. Corr = 40%

<http://data.giss.nasa.gov/modelforce/strataer/> Data: Global, NH, SH

VA := READPRN("Aerosols-Monthly-Mean Optical Thickness_tau_line.dat")



PMOD Solar Irradiance

TSI_{PMOD} := READPRN("TSI from 1979 to 2009-PMOD composite_d41_62_0906.txt")

TSI := READPRN("TSIpmod2.txt")

DateTSI := READPRN("TSIpmodDate.txt")

TSI_{Yr} := 1980 + floor($\text{TSI}_{\text{PMOD}}^{\langle 1 \rangle} \cdot 365^{-1}$)


TSIMon := ceil[(DateTSI - TSI_{Yr}) · 12]

TAM, Convert Daily to Monthly Average

TSIMonAvg := TAM(TSIMon, TSI_{Yr}, TSI)

TSI_{smm} := READPRN("TSIsm8.txt")

PMOD Total Solar Irradiance (Red), Smoothed (Yellow), Monthly (Blue)



Reconstruction of Solar Irradiance since 1610, Lean 1995 (1600-1995)


ftp://ftp.ncdc.noaa.gov/pub/data/paleo/contributions_by_author/lean1995/

TSD_{lean} := READPRN("TSDLeanFilled.txt")

TSD_{Flean} := READPRN("lean1995data.txt") Yr_{lean} := TSD_{Flean}^{⟨0⟩}

Solar Irradiance Correlates with U.S. Temp Anomaly


Lean Reconstructed Solar Irradiance (Red), PMOD +1.5 (Green), and Temp Anomaly (Blue)



Zonal Variation of Irradiance

<http://data.giss.nasa.gov/cgi-bin/cdrar/effij.py>

IrradZonal := READPRN("Irradiance zonal01.dat")



Monthly Time Series Matrices with Optimized Lags for Period 1980 - 2005

Let Y be the Temp and X1, X2, X3, and X4 be the delayed matrices for ENSO, Irradiance, Volcanic Aerosols, and Anthropogenic Influence.

HadCrut Temperature Data


HadCrut_{260,0} = 1980 2005 - 1980 = 25 r25 := 0..26 YrEven_{r25} := 260 + 2·r25 n := 0..12

HadCrutDat_{r25,n} := HadCrut_{260+r25·2,n} HadCrutMDat_R := rows(HadCrutDat) rr := 0..(R - 1)·12 + 11

HCTime_{rr} := (HadCrutDat⁽⁰⁾)_{floor(rr/12) + mod(rr,12)/12} HCTemp_{rr} := HadCrutDat_{floor(rr/12), mod(rr,12)+1}

Y := submatrix(HCTemp,0,12·26,0,0) Year_Y := submatrix(MEID,360,360 + 12·26,0,0) MEID₃₆₀ = 1980

HadCrut Monthly Global Temp (Bars) vs. Time



Empirical Component Data for 1980 to 2005, X1, X2, X3, X4, Given Optimized Lags (Months), Δt:
(The Optimization Procedure and Optimized Month Delays (3, 6, 0, 17Yrs) follows on pg 7.)

Lag Operator Δt(X, N=1980, τ): $\Delta t(X, N_{1980}, \tau) := \text{submatrix}(X, N_{1980} - \tau, N_{1980} - \tau + 12 \cdot 26, 0, 0)$

Define Forcing Parameters: X are normal forcings, Xs are Gasussian Smoothed

X1 = ENSO (E): 1980 through 2005 Δte = 3, MEID₁₉₈₀ = 1980 $x1 := \Delta t(\text{MEI}, 360, 3)$
 $\mu1 := \text{mean}(x1)$ $\sigma1 := \text{stderr}(\text{Year}_Y, x1)$ $z1 := (x1 - \mu1) \cdot \sigma1^{-1}$ $xs1 := \text{ksmooth}(\text{Year}_Y, z1, 2)$

X2 = Monthly Volcanic Aerosols (V), tv = 6, VA_{1560,0} = 1980.04
 $m := 0..100$ $\text{Zeros}_m := 0$ $\text{VAZ} := \text{stack}(\text{VA}^{\langle 1 \rangle}, \text{Zeros})$ $x2 := \Delta t(\text{VAZ}, 1560, 6)$
 $\mu2 := \text{mean}(x2)$ $\sigma2 := \text{stderr}(\text{Year}_Y, x2)$ $z2 := (x2 - \mu2) \cdot \sigma2^{-1}$ $xs2 := \text{ksmooth}(\text{Year}_Y, z2, 2)$

X3 = Solar Irradiance (S), ts = 0 $(\text{TSIMonAvg}^{\langle 0 \rangle})_{13} = 1980$
 $x3 := \Delta t(\text{TSIMonAvg}^{\langle 1 \rangle}, 13, 0)$ $\mu3 := \text{mean}(x3)$ $\sigma3 := \text{stderr}(\text{Year}_Y, x3)$ $z3 := (x3 - \mu3) \cdot \sigma3^{-1}$
 $xs3 := \text{ksmooth}(\text{Year}_Y, z3, 2)$

X4 = Anthropogenic Forcing, ΔF, of CO2 ppm (A), TA = 10*17 (17 Yr Delay)

Assume total radiative forcing (includes CH4) is proportional to radiative forcing due to carbon dioxide.

$\Delta F(C) := 4.841 \ln\left(\frac{C}{C_0}\right) + 0.0906 \cdot (\sqrt{C} - \sqrt{C_0})$ $\Delta F_{\text{IPCC}}(C) := 6.3 \ln\left(\frac{C}{C_0}\right)$
 $\text{CO2}_{\text{ppm}} := \Delta t(\text{TrendCO2}, 261, 17 \cdot 12)$ $x4 := \Delta F_{\text{IPCC}}(\text{CO2}_{\text{ppm}})$ $\text{Date}_{262} = 1980.042$
 $\mu4 := \text{mean}(x4)$ $\sigma4 := \text{stderr}(\text{Year}_Y, x4)$ $z4 := (x4 - \mu4) \cdot \sigma4^{-1}$ $xs4 := \text{ksmooth}(\text{Year}_Y, z4, 2)$

X5 = Pacific Decadal Oscillation (PDO) Index (PDO), TD = 10*17 PDOYr₉₆₀ = 1980

$x5 := \Delta t(\text{PDOI}, 960, 3)$ $\mu5 := \text{mean}(x5)$ $\sigma5 := \text{stderr}(\text{Year}_Y, x5)$ $z5 := (x5 - \mu5) \cdot \sigma5^{-1}$

Multi-Variate Component Construction and Design Matrix (Optimized Lags):

$$Y_i = \beta_0 + \beta_1 \cdot X1_i + \beta_2 \cdot X2_i + \beta_3 \cdot X3_i + \beta_4 \cdot X4_i + \epsilon_i$$

The design matrix for our Temp Stats data can be constructed with the components $i := 0..313 - 1$

ONE_i := 1 **ENSO** **Volcanic Aerosols** **Solar** **Anthro - Effects CO2**

$X^{\langle 0 \rangle} := \text{ONE}$ $X^{\langle 1 \rangle} := z1$ $X^{\langle 2 \rangle} := z2$ $X^{\langle 3 \rangle} := z3$ $X^{\langle 4 \rangle} := z4$

$Xs^{\langle 0 \rangle} := \text{ONE}$ $Xs^{\langle 1 \rangle} := xs1$ $Xs^{\langle 2 \rangle} := xs2$ $Xs^{\langle 3 \rangle} := xs3$ $Xs^{\langle 4 \rangle} := xs4$

$b := (X^T \cdot X)^{-1} \cdot (X^T \cdot Y)$ $b^T = (0.21959 \ 0.06608 \ -0.05037 \ 0.03063 \ 0.01255)$

$bs := (Xs^T \cdot Xs)^{-1} \cdot (Xs^T \cdot Y)$ $bs^T = (0.21978 \ 0.14923 \ -0.09635 \ 0.04577 \ 0.01226)$

Save Model β Coefficient Results: WRITEPRN("Emp_ESVA_Coefficients.txt") := b[■]

$\Delta T := b_0 + b_1 \cdot X^{\langle 1 \rangle} + b_2 \cdot X^{\langle 2 \rangle} + b_3 \cdot X^{\langle 3 \rangle} + b_4 \cdot X^{\langle 4 \rangle}$ $\Delta Ts := bs_0 + bs_1 \cdot xs1 + bs_2 \cdot xs2 + bs_3 \cdot xs3 + bs_4 \cdot xs4$

corr(Y, ΔT) = 0.87445

RSquare := corr(Y, ΔT)² = 0.76467

corr(Y, ΔTs)² = 0.77449

Optimization Procedure and Results:

Determine Time Lags to Maximize Correlation (R^2) of Regression Model to Global Temperature
 Evaluate lags of 0 to 12 Months for ENSO, Aerosols, and Irradiance and 5 to 20 yrs for Effects of CO2 ppm

```

OptLags(Y) :=
  X<0> ← ONE
  col ← MaxCorr ← 0
  for te ∈ 3..3
    for ta ∈ 6..6
      for ti ∈ 0..2
        for tc ∈ 15..20
          X<1> ← Δt(MEI, 360, te)
          X<2> ← Δt(VAZ, 1560, ta)
          X<3> ← Δt(TSIMonAvg<1>, 13, ti)
          co2ppm ← Δt(TrendCO2, 261, tc·12)
          X<4> ← ΔFIPCC(co2ppm)

          b ← (XT·X)-1·(XT·Y)
          ΔT ← b0 + b1·X<1> + b2·X<2> + b3·X<3> + b4·X<4>

          Corr ← corr(Y, ΔT)2
          if Corr > MaxCorr
            Opt1,0 ← col
            MaxCorr ← Corr
            Opt<col> ← (Corr te ta ti tc)T
            col ← col + 1
  Opt
  
```

Run Opt Routine and Gather Data

OptDat := OptLags(Y) OptDat_{1,0} = 2

OptDat^T := OptDat^T

OptDat^{<OptDat_{1,0}>} = (0.76467 3 6 0 17)

max(OptDat^{<0>}) = 0.76467

Optimization Results

For Optimum E, V, S, A Lags:
3, 6, 0, 17 (yrs), respectively.

$R^2 = 0.76$

The regression and estimation results are:


The numbers shown in parentheses below the regression coefficients are the magnitudes of their t-ratios; i.e. the coefficients divided by the standard deviation of the regression coefficient. All but the coefficient for Sunspot number are significantly different from zero at the 95 percent level of confidence and they are of the right sign.

Shown below is a comparison of the observed temperature change and the temperature change predicted by the regression equation. The observations are shown in red and the estimations from the regression equation are shown in blue.

Another way of viewing the comparison is in the scatter diagram below of the actual and regression predicted temperature changes.

The t-ratios for the variables included in the regression equation are significant. They **explain 76 percent of the variation in the year-to-year temperature change**. The insolation and CO2 ppm both a 76% correlation. Also the effect of the CO2 in the equation includes the effects of all variables influencing temperature change which are correlated with the general trend on CO2 concentration but are not in the equation. These would **include the effects of anthropogenic water vapor and anthropogenic cloudiness**.

Compare Anthropogenic Forcing Component - (17 Year Lag) = $b_4 \cdot z_4$ (Green) of ΔT to Data
 $b_4 \cdot z_4$ is the Optimized Match of Effects of Antro Forcing to Global Temperature Data




Statistical Analysis:

Four Factors - ENSO, Volcanic Aerosols, Insolation, and CO2 ppm explain 3/4 (76%) of the temperature variation.

int := intercept(Y, ΔT) $\hat{s} := \text{RSquare}$
 int = 0.05168 s = 0.76467 δT := int + s · Y

Scatter Plot: Linear Regression delta T vs Temp Data, Y



Test for Possible Regression

By extending this test to include p slope parameters $H_0: \beta_1 = \beta_2 = \beta_3 = \dots = \beta_p = 0$

we have the equivalent test for the possibility of a multiple regression, $H_0: \text{no multiple regression relationship}$

As in simple linear regression, we can associate each data value with three types of deviations, specifically, the residual error, e

$$\hat{y} := X \cdot b \qquad e := Y - \hat{y}$$

Sum of Squares

We can also use matrices to calculate the sum of squares for residual error,

$$SSE := e^T \cdot e \qquad SSE = 2.36546$$

as well as for regression, $SSR := \hat{y} - \text{mean}(Y)^T \cdot (\hat{y} - \text{mean}(Y))$

The total sum of squares equals n := rows(Y) p := 4

$$SST := SSE + SSR \qquad DF_REG := p \qquad DF_RESID := n - (p + 1) \qquad DF_RESID = 308$$

Mean Squares

Again, as in simple linear regression, dividing each sum of squares by the corresponding degrees of freedom provides us with variance estimates. The mean square for residual error

$$MSE := \frac{SSE}{DF_RESID}$$

$$MSR := \frac{SSR}{DF_REG}$$

$$DF_TOTAL := n - 1$$

F Test

The final entry in the table is the F statistic and corresponding p-value for the significance of an overall multiple regression. Under the null hypothesis of

H_0 : no regression relationship

$$\text{the test statistic } F := \frac{MSR}{MSE} \quad Rsq := \frac{SSR}{SST} \quad Rsq = 0.76467$$

has an F distribution with $n1 := DF_REG$ $n2 := DF_RESID$

degrees of freedom. The p-value of the test, then, is given by

$$p_val := 1 - pF(F, n1, n2) \quad p_val = 0$$

Summary: Analysis of Variance Table

Summarizing the above for our example,

DF	SS	MS	F
DF_REG = 4	SSR = 7.68624	MSR = 1.92156	F = 250.20084
DF_RESID = 308	SSE = 2.36546	MSE = 0.00768	p-value
DF_TOTAL = 312	SST = 10.0517		p_val = 0

The amount of variability explained by the linear regression (MSR) is greater than the amount due to residual error (MSE). The difference is large enough (the p-value is, in fact, close to 0) to strongly reject the null hypothesis,

Correlations between each pair of variables

in the model can be displayed in matrix form as

$$FLEX := \text{augment}(X^{(1)}, X^{(2)}, X^{(3)}, X^{(4)}, Y)$$

$$j := 0..p \quad k := 0..p$$

$$CORR_{j,k} := \text{corr}(FLEX^{(j)}, FLEX^{(k)})$$

STATISTICAL CORRELATIONS TO COMPONENTS

Correlation between "independent" variables

41% correlation between x1 (ENSO) and x2 (Volcanic)

Strong Correlation to Global Temp between:

Y (Temp) and x2 (Volcanic Aero) = 38%

Y (Temp) and x4 (Anthropogenic) = 78%

$$CORR = \begin{matrix} & \begin{matrix} x1 & x2 & x3 & x4 & y \end{matrix} \\ \begin{pmatrix} 1 & 0.41176 & -0.0859 & -0.10573 & 0.1549 \\ 0.41176 & 1 & -0.03022 & -0.31352 & -0.3816 \\ -0.0859 & -0.03022 & 1 & -0.15104 & 0.03587 \\ -0.10573 & -0.31352 & -0.15104 & 1 & 0.78122 \\ 0.1549 & -0.38169 & 0.03587 & 0.78122 & 1 \end{pmatrix} & \begin{pmatrix} x1 \\ x2 \\ x3 \\ x4 \\ y \end{pmatrix} \end{matrix}$$

Evaluate t tests

$$\text{Var_Covar_b} := (X^T \cdot X)^{-1} \cdot \text{MSE}$$

$$k := 0..p$$

$$\text{se_b}_k := \sqrt{\text{Var_Covar_b}_{k,k}}$$

$$\text{se_b}^T = (0.00495 \quad 0.00544 \quad 0.00549 \quad 0.00499 \quad 0.00049)$$

$$t := \frac{b}{\text{se_b}}$$

t tests

$$t^T = (44.3313 \quad 12.15164 \quad -9.18287 \quad 6.14094 \quad 25.55555)$$

2010 to 2020 Climate Forecasting:

Empirical Component Data and Forecast: X1, X2, X3, X4

X1 = ENSO: 1980 through 2010, then mimic 1992 to 1997 twice, $\Delta t_\epsilon = 3$

X2 = Monthly Volcanic Aerosols, repeat Pinatubo eruption with a peak in 2014, $\Delta t_v = 6$,

X3 = Irradiance: 1980 through 2009, then replicate cycle 23, $\Delta t_s = 0$:

X4 = Anthropogenic, $\Delta t_a = 17 \cdot 12$ (17 Yr Delay), then maintain trend.

Multi-Variate Component Forecast Model (β s Determined from Previous Regression):

$$Y_i = \beta_0 + \beta_1 \cdot X1_i + \beta_2 \cdot X2_i + \beta_3 \cdot X3_i + \beta_4 \cdot X4_i + \epsilon_i$$

The design matrix for our Temp Stats data can be constructed with the statements

$$X^{(1)} := x1 \quad X^{(2)} := x2 \quad X^{(3)} := x3 \quad X^{(4)} := x4$$

$$b := \text{READPRN}(\text{"Emp_ESVA_Coefficients.txt"})$$

$$\Delta T_{\text{forecast}} := b_0 + b_1 \cdot x1 + b_2 \cdot x2 + b_3 \cdot x3 + b_4 \cdot x4$$

The regression and estimation forecast:

